

Series Solutions of Kepler's Equation

Marc A. Murison

Astronomical Applications Department
U.S. Naval Observatory
Washington, DC

murison@aa.usno.navy.mil
<http://aa.usno.navy.mil/murison/>

June 26, 1998

Contents

- [Introduction](#)
- [An Iterative Method](#)
- [A Series Expansion Method](#)
- [A Fourier Sine Series Expansion and Resulting Bessel Function Representation for the Coefficients](#)
- [Chebyshev Polynomials](#)
- [A Chebyshev Series Expansion](#)

Introduction

Kepler's equation occurs in the context of the Newtonian two-body problem. The relative orbit of one body with respect to the other is easily characterized with the *true anomaly* as the independent variable. The true anomaly is just the angle: pericenter – focus — body, where focus is the ellipse focus around which the body moves. This is adequate for determining the orbit in space – its shape, size, and orientation. However, if one wishes to determine the orbit in time, things get more complicated, and one must solve Kepler's equation:

$$\phi = e \sin \phi + M$$

where $\phi(t)$ is the *eccentric anomaly*, e is the orbital eccentricity, $M = n(t - t_0)$ is the *mean anomaly*, t_0 is the time of pericenter passage, and n is the mean motion. The true and eccentric anomalies are related by

$$\cos v = \frac{\cos \phi - e}{1 - e \cos \phi}$$

and

$$\sin v = \frac{\sqrt{1 - e^2} \sin \phi}{1 - e \cos \phi}$$

Hence, to determine $v(t)$ we must first solve Kepler's equation for $\phi(t)$.


Since Kepler's equation is transcendental, we require iterative or series expansion approaches. Numerically, one can employ various fast algorithms. Here, we are not concerned with numerical methods but rather the analytical properties of the solution, so we will consider algebraic approximations.

[-] ellipse cartoon calculations

```

restart
with(plottools)
with(plots)
a := 5
b := 4
c := √(a² - b²)
e := c/a
θ_max := 15 π / 24
N := 25
x := array(1 .. N)
y := array(1 .. N)
for i to N do v := (i - 1) θ_max / (N - 1); r := a * (1 - e²) / (1 + e * cos(v)); x_i := r * cos(v) + c; y_i := r * sin(v) od
p1 := ellipse([0, 0], a, b, filled = true, color = yellow)
p2 := polygon([[c, 0], seq([x_i, y_i], i = 1 .. N), [c, 0]], color = red)
display(p2, p1, axes = normal, scaling = constrained)

```



[-] An Iterative Method

Make a first approximation, $\phi = M$, and substitute into the rhs of the Kepler equation $\phi = e \sin(\phi) + M$ to get a better approximation. Keep doing this. One finds the following succession of approximations:

```

restart
alias(phi = phi(e, M))
Keq := phi = e * sin(phi) + M
N := 6
phi[0]:=M;
for j to N do
  phi[j] := subs(phi=rhs(%),rhs(Keq));
  subs(f=phi, expansion(subs(phi=f,%), e, j));
  collect(combine(%,trig),e);
  print(%)
od:

```

$$\phi_0 = M$$

$$\phi_1 = e \sin(M) + M$$

$$\phi_2 = M + e \sin(M) + \frac{1}{2} e^2 \sin(2M)$$

$$\phi_3 = \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \frac{1}{2} e^2 \sin(2M) + e \sin(M) + M$$

$$\phi_4 =$$

$$\left(\frac{1}{3} \sin(4M) - \frac{1}{6} \sin(2M) \right) e^4 + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \frac{1}{2} e^2 \sin(2M) + e \sin(M) + M$$

$$\phi_5 = \left(-\frac{27}{128} \sin(3M) + \frac{1}{192} \sin(M) + \frac{125}{384} \sin(5M) \right) e^5 + \left(\frac{1}{3} \sin(4M) - \frac{1}{6} \sin(2M) \right) e^4 \\ + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \frac{1}{2} e^2 \sin(2M) + e \sin(M) + M$$

$$\phi_6 = \left(\frac{1}{48} \sin(2M) + \frac{27}{80} \sin(6M) - \frac{4}{15} \sin(4M) \right) e^6 \\ + \left(-\frac{27}{128} \sin(3M) + \frac{1}{192} \sin(M) + \frac{125}{384} \sin(5M) \right) e^5 + \left(\frac{1}{3} \sin(4M) - \frac{1}{6} \sin(2M) \right) e^4 \\ + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \frac{1}{2} e^2 \sin(2M) + e \sin(M) + M$$

A Series Expansion Method

Another approach, which yields the same result, is to construct a trial solution in the form of a power series in eccentricity, $\phi = \sum_{n=0}^N C_n e^n$

$$\phi = C_0 + C_1 e + C_2 e^2 + C_3 e^3 + C_4 e^4 + C_5 e^5 + C_6 e^6$$

Substitute this back into the Kepler equation and solve for the coefficients. Here is a Maple procedure that does this:

```
#-----
# Solve an equation phi = F(phi,eps) by series expansion, where
# phi is the solution variable and eps is a small parameter.
#-----
xsolve := proc( expr::{algebraic,algebraic=algebraic},
                 soln_var::name,function,
                 small_param::name,
                 expansion_order::posint )
```

```

local k, eqn, sols, S, trial, C;

if type(expr, `=`) then
  S := lhs(expr) - rhs(expr);
else
  S := expr;
fi;

#Create a trial solution of the form
# phi = C[0] + C[1]*eps + ... + C[N]*eps^N
#and substitute that into the equation, then
#expand into a power series to order N.
trial := sum( C[k]*small_param^k, k=0..expansion_order );
subs( soln_var=trial, S );
S := expansion( %, small_param, expansion_order );

#Solve for the coefficients C[k], starting with C[0]
#and successively working our way up to C[N], by equating
#coefficients of like powers of eps to zero.
sols := [];
for k from 0 to expansion_order do
  coeff( S, small_param, k );
  eqn := isolate( subs(sols,%), C[k] );
  if nargs > 4 then
    eqn := args[5]( eqn, args[6..nargs] );
  fi;
  sols := [ op(sols), eqn ];
od;

soln_var = subs( sols, trial );

end;

```

Applying this to the Kepler equation, we find

`xsolve(Keq, φ, e, N, combine, trig)`

$$\begin{aligned} \phi = M + \sin(M)e + \frac{1}{2}\sin(2M)e^2 + & \left(\frac{3}{8}\sin(3M) - \frac{1}{8}\sin(M)\right)e^3 \\ & + \left(-\frac{1}{6}\sin(2M) + \frac{1}{3}\sin(4M)\right)e^4 + \left(-\frac{27}{128}\sin(3M) + \frac{1}{192}\sin(M) + \frac{125}{384}\sin(5M)\right)e^5 \\ & + \left(\frac{1}{48}\sin(2M) + \frac{27}{80}\sin(6M) - \frac{4}{15}\sin(4M)\right)e^6 \end{aligned}$$

`φ6 := %`

Notice that we can regroup this as a Fourier series:

`collect(% sin)`

$$\begin{aligned} \phi = & \left(\frac{1}{48}e^6 + \frac{1}{2}e^2 - \frac{1}{6}e^4\right)\sin(2M) + \left(\frac{1}{3}e^4 - \frac{4}{15}e^6\right)\sin(4M) + \frac{125}{384}e^5 \sin(5M) \\ & + \left(-\frac{1}{8}e^3 + \frac{1}{192}e^5 + e\right)\sin(M) + \frac{27}{80}\sin(6M)e^6 + \left(\frac{3}{8}e^3 - \frac{27}{128}e^5\right)\sin(3M) + M \end{aligned}$$

- A Fourier Sine Series Expansion and Resulting Bessel Function Representation for the Coefficients

Suppose we expand $e \sin \phi$ in a Fourier series:

$$e \sin \phi = \sum_{k=1}^{\infty} 2 b_k \sin(k M)$$

where

$$\pi b_k = \int_0^{\pi} e \sin(\phi) \sin(k M) dM$$

Integrate this by parts to get

$$\pi b_k = \text{intparts(rhs(%), sin(phi))}$$

$$\pi b_k = -\frac{\sin(\phi(e, \pi)) e \cos(k \pi)}{k} + \frac{\sin(\phi(e, 0)) e}{k} - \int_0^{\pi} -\frac{\cos(\phi) \left(\frac{\partial}{\partial M} \phi \right) e \cos(k M)}{k} dM$$

The first two terms are zero, so we are left with $\pi b_k = \text{select(has, rhs(%), Int)}$:

$$\pi b_k = - \int_0^{\pi} -\frac{\cos(\phi) \left(\frac{\partial}{\partial M} \phi \right) e \cos(k M)}{k} dM$$

Now, $\left(\frac{\partial}{\partial M} \phi \right) e \cos(\phi) dM = d(e \sin(\phi))$. Hence, since we can also write $e \sin \phi = \phi - M$, we have $d(e \sin(\phi)) = d\phi - dM$ and

$$\pi b_k = \int_0^{\pi} \frac{\cos(k M)}{k} d\phi - \int_0^{\pi} \frac{\cos(k M)}{k} dM$$

The second integral is zero for integer $k \neq 0$:

`Int(cos(k*M)/k, M=0..Pi) : % = value(%);`

$$\int_0^{\pi} \frac{\cos(kM)}{k} dM = \frac{\sin(k\pi)}{k^2}$$

Thus, we are left with

$$\pi b_k = \int_0^{\pi} \frac{\cos(k(\phi - e \sin(\phi)))}{k} d\phi$$

Now, the Bessel function of the first kind is defined as

$$\pi J_n(x) = \int_0^{\pi} \cos(n\theta - x \sin(\theta)) d\theta$$

where n is an integer. Therefore, finally, we have

$$b_k = \frac{J_k(k e)}{k}$$

and

$$\phi = M + 2 \left(\sum_{k=1}^{\infty} \frac{J_k(k e) \sin(k M)}{k} \right)$$

Let's compare this with our previous series expansion example, phi6

$$\begin{aligned} \phi = & \left(\frac{1}{48} \sin(2M) + \frac{27}{80} \sin(6M) - \frac{4}{15} \sin(4M) \right) e^6 \\ & + \left(-\frac{27}{128} \sin(3M) + \frac{1}{192} \sin(M) + \frac{125}{384} \sin(5M) \right) e^5 + \left(\frac{1}{3} \sin(4M) - \frac{1}{6} \sin(2M) \right) e^4 \\ & + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \frac{1}{2} e^2 \sin(2M) + e \sin(M) + M \end{aligned}$$

$$M + 2 \left(\sum_{k=1}^N \frac{J_k(k e) \sin(k M)}{k} \right)$$

series(subs(seq(J_k(k e) = BesselJ(k, k e), k = 1 .. N), %), e, N+1)

$$\begin{aligned} & M + \sin(M)e + \frac{1}{2} \sin(2M)e^2 + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 + \left(\frac{1}{3} \sin(4M) - \frac{1}{6} \sin(2M) \right) e^4 \\ & + \left(-\frac{27}{128} \sin(3M) + \frac{1}{192} \sin(M) + \frac{125}{384} \sin(5M) \right) e^5 + \end{aligned}$$

$$\left(\frac{1}{48} \sin(2M) + \frac{27}{80} \sin(6M) - \frac{4}{15} \sin(4M) \right) e^6 + O(e^7)$$

simplify(convert(% , polynom) - rhs(phi6))

0

Since

$$1 = J_0(x)^2 + 2 \left(\sum_{k=1}^{\infty} J_k(x)^2 \right)$$

we know that the series

$$\frac{\phi - M}{2} = \sum_{k=1}^{\infty} \frac{J_k(k e) \sin(k M)}{k}$$

is *absolutely* convergent.

- Chebyshev Polynomials

- Definition

Suppose we expand a function $f(x)$ in a Chebyshev series:

$$f(x) = \sum_{k=0}^{\infty} a_k T_k(x)$$

The Chebyshev polynomials of the first kind are defined

$$T_k(x) = \cos(k\theta)$$

where $x = \cos(\theta)$. An explicit polynomial representation is

$$T_n(x) = \sum_{k=0}^{[n/2]} \frac{(-1)^k (n-k-1)! (2x)^{(n-2k)}}{k! (n-2k)!}$$

For example, the first several Chebyshev polynomials are

$$\text{for } j \text{ to 12 do value} \left(\text{subs} \left(n=j, T_n(x) = \sum_{k=0}^{\frac{n}{2}} \frac{n(-1)^k (n-k-1)! (2x)^{(n-2k)}}{2k! (n-2k)!} \right) \right) \text{od}$$

$$T_1(x) = x$$

$$T_2(x) = 2x^2 - 1$$

$$T_3(x) = 4x^3 - 3x$$

$$T_4(x) = 8x^4 - 8x^2 + 1$$

$$T_5(x) = 16x^5 - 20x^3 + 5x$$

$$T_6(x) = 32x^6 - 48x^4 + 18x^2 - 1$$

$$T_7(x) = 64x^7 - 112x^5 + 56x^3 - 7x$$

$$T_8(x) = 128x^8 - 256x^6 + 160x^4 - 32x^2 + 1$$

$$T_9(x) = 256x^9 - 576x^7 + 432x^5 - 120x^3 + 9x$$

$$T_{10}(x) = 512x^{10} - 1280x^8 + 1120x^6 - 400x^4 + 50x^2 - 1$$

$$T_{11}(x) = 1024x^{11} - 2816x^9 + 2816x^7 - 1232x^5 + 220x^3 - 11x$$

$$T_{12}(x) = 2048x^{12} - 6144x^{10} + 6912x^8 - 3584x^6 + 840x^4 - 72x^2 + 1$$

Chebyshev polynomials satisfy the following differential equations:

$$(1-x^2) \left(\frac{\partial^2}{\partial x \partial x} y(x) \right) - x \left(\frac{\partial}{\partial x} y(x) \right) + k^2 y(x) = 0 \text{ for } T_k(x)$$

and


$$\left(\frac{\partial^2}{\partial \theta \partial \theta} y(\theta) \right) + k^2 y(\theta) = 0 \text{ for } T_k(\cos \theta)$$

Here is a graphical view of the first seven polynomials:

with(*polynomials*)

plotpoly([1, 2, 3, 4, 5, 6, 7], *T*)

Chebyshev type 1



- A Recurrence Relation

From the definition $T_k(x) = \cos(k\theta)$, we have

```
cos((k+1)*theta) : % = expand(%);  
cos((k+1)*theta) = cos(theta*k)*cos(theta) - sin(theta*k)*sin(theta)  
or,  
subs(A = sin(theta*k)*sin(theta), isolate(subs(sin(theta*k)*sin(theta) = A, %), A))  
sin(theta*k)*sin(theta) = -cos((k+1)*theta) + cos(theta*k)*cos(theta)
```

We also have

```
cos((k+2)*theta) : % = expand(%);  
cos((k+2)*theta) = 2*cos(theta*k)*cos(theta)^2 - cos(theta*k) - 2*sin(theta*k)*sin(theta)*cos(theta)  
or, substituting for sin(theta*k)*sin(theta),  
algsubs(%%%, %)  
cos((k+2)*theta) = -cos(theta*k) + 2*cos(theta)*cos((k+1)*theta)
```

From this we have the useful recurrence relation

$$T_{k+2} = 2xT_{k+1} - T_k$$

- Orthogonality

The Chebyshev polynomials are orthogonal:

$$\int_0^\pi T_n(\cos \theta) T_m(\cos \theta) d\theta = 0 \text{ for } n \neq m$$

and

$$\int_0^\pi T_n(\cos \theta)^2 d\theta = \begin{cases} \frac{\pi}{2} & \text{for } n=0 \\ 0 & \text{for } n \neq 0 \end{cases}$$

If we write $x = \cos \theta$, these become

$$\int_{-1}^1 \frac{T_n(x) T_m(x)}{\sqrt{1-x^2}} dx = 0 \text{ for } n \neq m$$

and

$$\int_{-1}^1 \frac{T_n(x)^2}{\sqrt{1-x^2}} dx = \begin{cases} \frac{\pi}{2} & \text{for } n=0 \\ 0 & \text{for } n \neq 0 \end{cases}$$

- Least Squares Fit and Determination of Chebyshev Series Coefficients

Let us write the truncation error for an order N approximation of $f(x)$:

$$\epsilon_N = f(x) - \left(\sum_{k=0}^N a_k T_k(x) \right)$$

One measure of goodness of fit of a polynomial to a function is the least squares integral

$$S_N = \int_{-1}^1 w(x) \epsilon_N^2 dx$$

for some weighting function $w(x)$. To minimize S_N we set $\frac{\partial}{\partial a_k} S_N = 0$ for all the a_k , finding

$$\int_{-1}^1 2 w(x) \left(f(x) - \left(\sum_{n=0}^N a_n T_n(x) \right) \right) T_k(x) dx = 0$$

Let the weighting function be $w(x) = \frac{1}{\sqrt{1-x^2}}$. Then

$$\sum_{n=0}^N a_n \int_{-1}^1 \frac{T_n(x) T_k(x)}{\sqrt{1-x^2}} dx = \int_{-1}^1 \frac{f(x) T_k(x)}{\sqrt{1-x^2}} dx$$

Using the orthogonality relations, we arrive at the results

$$\pi a_0 = \int_{-1}^1 \frac{f(x)}{\sqrt{1-x^2}} dx$$

and

$$\pi a_k = 2 \int_{-1}^1 \frac{f(x) T_k(x)}{\sqrt{1-x^2}} dx \text{ for } k \neq 0$$

These are equivalent to

$$\pi a_0 = \int_0^\pi f(\cos \theta) d\theta$$

and

$$\pi a_k = 2 \int_0^\pi f(\cos \theta) \cos(k \theta) d\theta \text{ for } k \neq 0$$

where $\cos \theta = x$.

A Chebyshev Series Expansion

Determination of the Coefficients

Suppose we expand $\phi - M = e \sin \phi$ in a Chebyshev series:

$$e \sin \phi = \sum_{k=1}^{\infty} 2 a_k T_k(\cos M)$$

The coefficients are

$$\pi a_k = \int_0^\pi e \sin(\phi) \cos(k M) dM$$

Integrate this by parts to get

$$\pi a_k = \text{intparts(rhs(%), sin(phi))}$$

$$\pi a_k = \frac{\sin(\phi(e, \pi)) e \sin(k \pi)}{k} - \int_0^\pi \frac{\cos(\phi) \left(\frac{\partial}{\partial M} \phi \right) e \sin(k M)}{k} dM$$

The first term is zero, so we are left with $\pi a_k = \text{select(has, rhs(%), Int)}$

$$\pi a_k = - \int_0^\pi \frac{\cos(\phi) \left(\frac{\partial}{\partial M} \phi \right) e \sin(k M)}{k} dM$$

Again, $\left(\frac{\partial}{\partial M} \phi \right) e \cos(\phi) dM = d(e \sin(\phi))$. Since we can write $e \sin \phi = \phi - M$, we again have $d(e \sin(\phi)) = d\phi - dM$ and, therefore,

$$\pi a_k = \int_0^\pi \frac{\sin(k M)}{k} dM - \int_0^\pi \frac{\sin(k M)}{k} d\phi$$

The first integral is

`Int(sin(k*M)/k, M=0..Pi) : % = value(%);`

$$\int_0^\pi \frac{\sin(k M)}{k} dM = - \frac{\cos(k \pi) - 1}{k^2}$$

Hence, we have

$$\pi a_k = \frac{1 - \cos(k \pi)}{k^2} - \int_0^\pi \frac{\sin(k(\phi - e \sin(\phi)))}{k} d\phi$$

Now, the Weber function is defined as

$$\pi E_v(x) = \int_0^\pi \sin(v\theta - x \sin(\theta)) d\theta$$

Therefore, we have

$$a_k = \frac{1 - \cos(k\pi)}{\pi k^2} - \frac{E_k(k)e}{k}$$

and

$$\phi = M + \left(\sum_{k=1}^{\infty} 2 \left(\frac{1 - \cos(k\pi)}{\pi k^2} - \frac{E_k(k)e}{k} \right) T_k(\cos M) \right)$$

The Weber Function

The first few terms of the series expansion of $E_k(x)$ are

$$E_k(x) = \text{series}(WeberE(k, x), x, 8)$$

$$\begin{aligned} E_k(x) = & \frac{\sin\left(\frac{1}{2}k\pi\right)}{\Gamma\left(1-\frac{1}{2}k\right)\Gamma\left(1+\frac{1}{2}k\right)} + \frac{1}{2} \frac{\cos\left(\frac{1}{2}k\pi\right)}{\Gamma\left(\frac{3}{2}-\frac{1}{2}k\right)\Gamma\left(\frac{3}{2}+\frac{1}{2}k\right)} x - \frac{1}{4} \frac{\sin\left(\frac{1}{2}k\pi\right)}{\Gamma\left(2-\frac{1}{2}k\right)\Gamma\left(2+\frac{1}{2}k\right)} \\ & x^2 - \frac{1}{8} \frac{\cos\left(\frac{1}{2}k\pi\right)}{\Gamma\left(\frac{5}{2}-\frac{1}{2}k\right)\Gamma\left(\frac{5}{2}+\frac{1}{2}k\right)} x^3 + \frac{1}{16} \frac{\sin\left(\frac{1}{2}k\pi\right)}{\Gamma\left(3-\frac{1}{2}k\right)\Gamma\left(3+\frac{1}{2}k\right)} x^4 + \\ & \frac{1}{32} \frac{\cos\left(\frac{1}{2}k\pi\right)}{\Gamma\left(\frac{7}{2}-\frac{1}{2}k\right)\Gamma\left(\frac{7}{2}+\frac{1}{2}k\right)} x^5 - \frac{1}{64} \frac{\sin\left(\frac{1}{2}k\pi\right)}{\Gamma\left(4-\frac{1}{2}k\right)\Gamma\left(4+\frac{1}{2}k\right)} x^6 - \\ & \frac{1}{128} \frac{\cos\left(\frac{1}{2}k\pi\right)}{\Gamma\left(\frac{9}{2}-\frac{1}{2}k\right)\Gamma\left(\frac{9}{2}+\frac{1}{2}k\right)} x^7 + O(x^8) \end{aligned}$$

$$\lim_{k \rightarrow 6} \text{convert(rhs(%), polynom)}$$

$$\frac{2}{135135} \frac{x(-143x^2 - 13x^4 - 3861 + x^6)}{\pi}$$

Compare with the definition:

$$\begin{aligned}
& \text{series} \left(\frac{\int_0^{\pi} \sin(v\theta - x \sin(\theta)) d\theta}{\pi}, x, 8 \right) \\
& - \frac{\cos(\pi v) - 1}{v \pi} + \frac{\cos(\pi v) + 1}{(v+1)(v-1)\pi} x - \frac{\cos(\pi v) - 1}{v(v+2)(v-2)\pi} x^2 + \\
& \frac{\cos(\pi v) + 1}{(v+3)(v-3)(v+1)(v-1)\pi} x^3 - \frac{\cos(\pi v) - 1}{v(v+4)(v-4)(v+2)(v-2)\pi} x^4 + \\
& \frac{\cos(\pi v) + 1}{(v+5)(v-5)(v+3)(v-3)(v+1)(v-1)\pi} x^5 - \\
& \frac{\cos(\pi v) - 1}{v(v+6)(v-6)(v+4)(v-4)(v+2)(v-2)\pi} x^6 + \\
& \frac{\cos(\pi v) + 1}{(v+7)(v-1)(v-7)(v+5)(v-5)(v+3)(v-3)(v+1)\pi} x^7 + O(x^8)
\end{aligned}$$

$\lim_{v \rightarrow 6}$ convert(% , polynom)

$$-\frac{2}{135135} \frac{x(-143x^2 - 13x^4 - 3861 + x^6)}{\pi}$$


It appears Maple's definition differs by a minus sign, so we'll define our own Weber function.

$$W := \text{fn} \left(\frac{\int_0^{\pi} \sin(v\theta - x \sin(\theta)) d\theta}{\pi}, v, x \right)$$

Compare the two functions:

`plot3d(W(3, x) - WeberE(3, x), x = -1 .. 1, k = 1 .. 10, grid = [30, 30], orientation = [-125, 60], shading = xyz, lightmodel = light3, title = "W(k,x) - WeberE(k,x)")`

W(k,x) - WeberE(k,x)



Aha, this is just numerical noise, so it's the Maple series expansion for Weber functions, procedure 'series/WeberE', that has the minus sign error, not the Weber function itself.

Comparison to the Taylor Series Expansion

Let's compare the Chebyshev expansion for ϕ with our previous series expansion example, ϕ_6

$$\begin{aligned}\phi = & M + \sin(M) e + \frac{1}{2} \sin(2M) e^2 + \left(\frac{3}{8} \sin(3M) - \frac{1}{8} \sin(M) \right) e^3 \\ & + \left(-\frac{1}{6} \sin(2M) + \frac{1}{3} \sin(4M) \right) e^4 + \left(-\frac{27}{128} \sin(3M) + \frac{1}{192} \sin(M) + \frac{125}{384} \sin(5M) \right) e^5 \\ & + \left(\frac{1}{48} \sin(2M) + \frac{27}{80} \sin(6M) - \frac{4}{15} \sin(4M) \right) e^6\end{aligned}$$

$f_taylor := \text{fn}(\text{rhs}(\phi_6), e, M)$

$$M + \left(\sum_{k=1}^N 2 \left(\frac{1 - \cos(k\pi)}{\pi k^2} - \frac{E_k(k e)}{k} \right) \text{orthopoly}_T(k, \cos(M)) \right)$$

$\text{subs}(\text{seq}(E_k(k e) = \text{WeberE}(k, k e), k = 1 .. N), \%)$

$f_weber := \text{fn}(\%, e, M)$

```

subs(seq( $E_k(k e) = W(k, k e)$ ,  $k = 1 .. N$ ), %%%)
expansion(% , e, N)
collect(%, e, combine)


$$\frac{1}{51975} \frac{(132 \cos(M) - 17820 \cos(3M) + 62500 \cos(5M)) e^6}{\pi}$$


$$+ \frac{1}{10395} \frac{(-2112 \cos(2M) + 11264 \cos(4M) - 5184 \cos(6M)) e^5}{\pi}$$


$$+ \frac{1}{945} \frac{(-84 \cos(M) + 972 \cos(3M) - 500 \cos(5M)) e^4}{\pi}$$



$$+ \frac{1}{105} \frac{(112 \cos(2M) - 64 \cos(4M) - 16 \cos(6M)) e^3}{\pi}$$


$$+ \frac{1}{105} \frac{(-20 \cos(5M) + 140 \cos(M) - 84 \cos(3M)) e^2}{\pi}$$


$$+ \frac{1}{105} \frac{(-140 \cos(2M) - 12 \cos(6M) - 28 \cos(4M)) e}{\pi} + M$$


f_weberx := fn(% , e, M)
Kepler := proc(e, M) local E; option remember; fsolve(E - e*sin(E) = M, E) end
e0 := .8
plot([0, f_taylor(e0, M) - 'Kepler(e0, M)', f_weber(e0, M) - 'Kepler(e0, M)',
f_weberx(e0, M) - 'Kepler(e0, M)'], M = 0 .. 2π, color = [black, blue, red, green])

```



Hmph. For some reason, the Chebyshev series approximation stinks.

A Fancy 3D View of the Error

```
[ with(plots)
[ p1 := tubeplot([cos(M), sin(M), 0], radius = f_taylor(e0, M) - 'Kepler(e0, M)', M = 0 .. 2 π,
[ color = blue, numpoints = 100, tubepoints = 50, style = patchnogrid)
[ p2 := tubeplot([cos(M), sin(M), 0], radius = f_weber(e0, M) - 'Kepler(e0, M)', M = 0 .. 2 π,
[ color = red, numpoints = 100, tubepoints = 50, style = wireframe)
[ p3 := tubeplot([cos(M), sin(M), 0], radius = f_weberx(e0, M) - 'Kepler(e0, M)', M = 0 .. 2 π,
[ color = green, numpoints = 100, tubepoints = 50, style = wireframe)
[ display(p1, p2, p3, orientation = [35, 40], lightmodel = light2, labels = ["cos M", "sin M", ""])]
```

